
Abstract

This paper presents work on methodologies for the programming tool GCLA II1. Two methods are

investigated for reasoning about properties of regular expressions and their corresponding nondeter-

ministic finite automata (NFA’s). The methods described are combination of predicate and functional

programming within the framework of GCLA II and an investigation of properties and possible usages

of the duality of the "," (and) construct within the language.

1 Introduction

This paper is the first result of a project currently carried out at the Department of Computer Sciences
the supervision of Lars Hallnäs. Within this project we intend to study the problem of giving mathemat
precise and implementation independent models of the notion of a knowledge base (KB). These mod
then serve as a basis for computer aided development and testing of specialized knowledge based
(KBS). We will especially concentrate on trying to characterize the logical interface of a KB, i.e. a u
basic conceptual model of a given KB. We will in particular look at applications in the field of diagno
error detection.

An example of such an application is environments for program development, especially debuggin
aim is to investigate how knowledge based systems (KBS) technology can be applied to such applic
using the programming tool GCLA II. The language GCLA II was originally designed for application
within KBS such as default and hypothetical reasoning, diagnosis and simulation.

The examples in this paper are taken from the area of lexical analysis, since this seems to be a ne
first step.

This paper contributes to the as yet poorly known domain of programming methodology for GCL
Two methods are investigated for reasoning about properties of regular expressions and their corresp
nondeterministic finite automata (NFA’s). The methods described are combination of predicate and
tional programming within the framework of GCLA II and an investigation of properties and possible
usages of the duality of the "," (and) construct within the language.

2 GCLA II

2.1 Introduction to GCLA II

GCLA2 II (Generalized HornClauseLAnguage) is perhaps best described as a logical programming
guage, with some properties usually found among functional languages.

Compared to Prolog, what has been added to GCLA II is the possibility to assume conditions. For
ple, the clause

a <= (b -> c).

1. This work was carried out as part of the work in the ESPRIT working group GENTZEN.

2. To be pronounced “gisela”.

should be read as “a holds ifc  can be proved while assumingb”.
There is also a richer set of queries in GCLA II than in Prolog. An ordinary Prolog query is writte

:- a.

and should be read as “Doesa hold (in the definitionD)?” In GCLA II one can also assume things in th
query, for example

c \- a.

which should be read as “Assumingc , doesa hold (in the definitionD)?”, or “Is a derivable fromc?”
To execute a program, a queryG is posed to the system, asking whether there is a substitutionσ such that

Gσ holds according to the logic defined by the program. The goalG has the formΓ |− c, whereΓ is a list of
assumptions, and c is the conclusion from the assumptionsΓ. The system tries to construct a deduction
showing thatGσ holds in the given logic.

For a more complete and comprehensive description of GCLA II’s theoretical properties see [1]. [2]
onstrates the program development methodology used in GCLA II. Various implementation techniq
including functional and object oriented programming, are also demonstrated.

2.2 Programs in GCLA II

A GCLA II program consists of two parts; One part is used to express the declarative content of the pro
called thedefinitionor theobject level, and the other part is used to express rules and strategies actin
the declarative part, called therule definition or themeta level.

2.2.1 The definition

The definition constitutes the formalization of a specific problem domain, and in general contains a
mum of control information. The intention is that the definition by itself gives a purely declarative des
tion of the problem domain while a procedural interpretation of the definition is obtained only by putt
in the context of the rule definition.

2.2.2 The rule definition

The rule definition contains the procedural knowledge of the domain, i.e., the knowledge used for dr
conclusions based on the declarative knowledge in the definition. This procedural knowledge defin
possible inferences made from the declarative knowledge.

The rule definition containsinference ruledefinitions, which define how different inference rules shou
act, and strategies which control the search among the inference rules.

2.2.3 Example: Default reasoning

Assume we know that an object can fly if it is a bird and if it is not a penguin. We also know that Tw
and Polly are birds as well as all penguins, and finally we know that Pengo is a penguin. This knowle
expressed in the following definition:

flies(X) <=
bird(X),
(penguin(X) -> false).

bird(tweety).
bird(polly).
bird(X) <=

penguin(X).

penguin(pengo).

To be able to use this definition the way we want we have to write the following rule definition:

fs <= % Never do axiom!
right(fs), % First try standard right rules.
left_if_false(fs). % else if consequent is false...

left_if_false(PT) <= % Is the consequent false?
(_ \- false).

left_if_false(PT) <= % If so perform left rules.
no_false_assump(PT),
false_left(_).

no_false_assump(PT) <= % No false assumption.
not(member(false,A)) -> % i.e. the term false is not a
(A \- _). % member of the assumption list.

no_false_assump(PT) <=
left(PT).

member(X,[X|_]). % Proviso definition.
member(X,[_|R]):-

member(X,R).

If we want to know which birds can fly, we pose the query

fs \\- (\- flies(X)).

and the system will respond withX = tweety  andX = polly .
If we want to know which birds cannot fly, we can pose the query

fs \\- (flies(X) \- false).

and the system will respond withX = pengo .

3 Some background material

In the following subsections we will give a short introduction tonondeterministic finite automataand
Thompson’s construction. For a more complete description of these subjects we refer to [3].

3.1 Nondeterministic finite automata

Definition: A nondeterministic finite automaton (NFA) is a mathematical model that consists of:

1. a set of statesS

2. a set of input symbolsΣ

3. a transition functionmove that maps state-symbol pairs to sets of state

4. a states0 that is distinguished as thestart state

5. a set of statesF distinguished asaccepting states

An NFA can be represented diagrammatically by a labelled directed graph, called a transition gra
which the nodes are the states and the labelled edges represent the transition function.

An NFA accepts an input strings if and only if there is some path in the transition graph correspondi
to the NFA from the start state to some accepting state, such that the edge labels along this path spes.
The language defined by an NFA is the set of input strings that it accepts.

3.2 Thompson’s construction

Given a regular expressionr, we first parser into its constituent subexpressions. Then using rule (1) and
below, we construct NFA’s for each of the basic symbols inr (those that are eitherε or an alphabet symbol).
Then, guided by the syntactic structure of the regular expressionr, we combine these NFA’s inductively
using rule (3) below until we obtain the NFA for the entire expression.

Each intermediate NFA produced during the course of the construction corresponds to a subexp
of r and has several important properties: it has exactly one final state, no edges enters the start st
no edge leaves the final state.

1. For the regular expressionε, construct the NFA

Here i is a new start state andf a new accepting state.

2. For the regular expressiona, construct the NFA

Again i is a new start state andf a new accepting state.

3. SupposeN(s) andN(t) are NFA’s for the regular expressionss andt respectively.

a) For the regular expressions|t, construct the composite NFA

Herei is a new start state andf a new accepting state.

b) For the regular expressionst, construct the composite NFA

The start state ofN(s) becomes the start state of the composite NFA and the accepting state oN(t)
becomes the accepting state of the composite NFA. The accepting state ofN(s) is merged with the
start state ofN(t).

c) For the regular expressions*, construct the composite NFA

Herei is a new start state andf a new accepting state.

d) For the parenthesized regular expression (s), useN(s) itself as the NFA.

e) For the regular expressions+, use the NFA of the expressionss*.

f) For the regular expressions?, use the NFA of the expressionε|s.

4 Mixing functions and predicates in programs

This section describes how we simulate an NFA in GCLA II using the possibility to mix functions and p
icates in a program.

start ε
fi

start a
fi

start i f

N(s)

N(t)

ε

ε

ε

ε

start i f
N(s) N(t)

start i fN(s) εε

ε

4.1 A functional version of Thompson’s construction

The algorithm described in Section 3.2 for constructing NFA’s is easily changed into a function that ta
regular expression as its argument and returns the corresponding NFA generated by Thompson’s c
tion. An NFA is represented by a 3-tuple consisting of a list, containing all the transitions of the NFA
start state and its accepting state. When we use a function in a query we put the function call to the
the turnstile, '\- ', and get the result to the right. Using the strategythomp/0 , which assures that we will
only get the answers we are interested in, we get the NFA corresponding to a regular expression li

thomp \\- thompson("a":"b") \- NFA.

NFA = nfa([m(0,"a",1),m(1,"b",2)],0,2)

The query above should be read: “What is the NFA corresponding to the regular expressionab?”. We could
also instantiate the result in which case the query should be read:
nfa([m(0,"a"],1),m(1,"b",2)],0,2) the NFA given by Thompson’s construction for the regul
expressionab.

The definition ofthompson/1 is simply a call tothompson/2 , which has one clause for each kind o
regular expression. We show how some of these should be read and then give the entire definition
thompson/2 .

thompson(eps,S) <=
(S+1 -> S1) -> nfa([m(S,eps,S1)],S,S1).

The clause above, which gives the NFA for the empty string, should be read: “IfS+1 evaluates toS1 then
thompson(1,eps,S)  evaluates tonfa([m(S,eps,S1)],S,S1) .”

thompson((A:B),S) <=
(thompson(A,S) -> nfa(M1,S,T2)),
(thompson(B,T2) -> nfa(M2,T2,T)),
append(M1,M2,M3)
-> nfa(M3,S,T).

This clause gives the NFA for the regular expressionsab. Here we have mixed functional and predicate pr
gramming, whereasappend/3 is a predicate that appends its first and second argument in the third.
way to read this clause is: “Ifthompson(A,S) evaluates to the NFAnfa(M1,S,T2) , and thomp-
son(B,T2) evaluates tonfa(M2,T2,T) , andM1andM2are appended inM3thenthompson((A:B),S)
evaluates to the NFAnfa(M3,S,T) .”

In the definition below, each clause corresponds to the different cases in the description of Thom
construction given in Section 3.2.

thompson(eps,S) <=
(S+1 -> S1) -> nfa([m(S,eps,S1)],S,S1).

thompson(A,S)
#{A\=(_|_),A\=(_:_),A\=eps,A\='?'(_),A\='+'(_),A\='*'(_),A\= reg(_,_)} <=

(S+1 -> S1) -> nfa([m(S,A,S1)],S,S1).

thompson((A|B),S) <=
(S+1 -> S1),(thompson(A,S1) -> nfa(M1,S1,T2)),
(T2+1 -> ST2),(thompson(B,ST2) -> nfa(M2,ST2,T)
(T+1 -> T1),
append(M1,M2,M3)
-> nfa([m(S,eps,S1),m(S,eps,ST2),m(T2,eps,T1),m(T,eps,T1)|M3],S,T1).

thompson((A:B),S) <=
(thompson(A,S) -> nfa(M1,S,T2)),
(thompson(B,T2) -> nfa(M2,T2,T)),
append(M1,M2,M3)
-> nfa(M3,S,T).

thompson((A)*,S) <=
(S+1 -> S1),(thompson(A,S1) -> nfa(M1,S1,T)),
(T+1 -> T1)
-> nfa([m(S,eps,T1),m(S,eps,S1),m(T,eps,T1),m(T,eps,S1)|M1],S,T1).

thompson((A)+,S) <=
thompson((A:A*),S).

thompson((A)?,S) <=
thompson((A|eps),S).

thompson(reg(X,E),S) <=
reg(X,E) -> thompson(E,S).

4.2 Running an NFA given by Thompson’s construction

To be able to determinate whether a certain string is recognized by a given NFA or not, we need to si
the NFA. For this purpose we use Algorithm 3.4 in [3].

An example of a typical query is:

sim \\- input("ab") \- simula(nfa([m(0,"a",1),m(1,"b",2)],0,2)).

This should be read: “Does the given NFA recognise the string"ab" ?”, and will give the answeryes . When
writing a definition in GCLA II you have to decide whether you want to use the definition on the left o
right side of the sequent or on both. When we wrote the definition ofsimula/1 we decided that it should
be used to the right, and that we did not want to give the input string as a parameter to the simulatio
rithm. Instead we use the left hand as a kind of storage and read characters from it when needed.

The simulation of an NFA uses two operationsmoveandε-closure. If S is a set of states thenmove(S,a)
gives all states that can be reached from a state inS by a transition ona, the current input character.ε-clo-
sure(S) gives all the states that can be reached from a state inS by zero or moreε-transitions.

We describe Algorithm 3.4 by showing its definition in GCLA II:

simula(nfa(Transes,S0,Final)) <=
simula(eclosure(stack([S0]),[S0],Transes),Transes,Final).

The first thing to do is to compute theε-closureof the start stateS0, which is done by a call to the function
eclosure/3 , execution then continues with a call tosimula/3 :

simula(S,Transes,Final) <=
next_char(A,(eq(A,[]),
member(Final,S))).

The above could be read: “If the next input characterA is the empty string, then see if the final state,Final ,
is in S, the current set of states”. If this is the case then the NFA accepts the given string.

The definition below tells us what to do if the next input character is not the empty string. First, a
states that can be reached by transitions on the current input characterA are computed bymove/4 and then
theε-closure of these are computed and execution continues with a call tosimula/3 .

simula(S,Transes,Final)<=
next_char(A,(neq(A,[]),
move(S,[A],Transes,Moves),
simula(eclosure(stack(Moves),Moves,Transes),Transes,Final))).

The entire definition ofsimula/1  andsimula/3  is

simula(nfa(Transes,S0,Final)) <=
simula(eclosure(stack([S0]),[S0],Transes),Transes,Final).

simula(E) #{E \= nfa(_,_,_)}<=
(E -> NFA),
simula(NFA).

simula(S,Transes,Final)#{S \= [], S \= [_|_]} <=
(S -> S1),
simula(S1,Transes,Final).

simula(S,Transes,Final) <=
next_char(A,(eq(A,[]),
member(Final,S))).

simula(S,Transes,Final)<=
next_char(A,(neq(A,[]),
move(S,[A],Transes,Moves),
simula(eclosure(stack(Moves),Moves,Transes),Transes,Final))).



-

d check

he

g-

, which
4.3 Some queries

Does the given NFA accept the input string?

sim \\- input("aaaaa") \-
simula(nfa([m(0,eps,3),m(0,eps,1),m(2,eps,3),
m(2,eps,1),m(1,"a"],2)],0,3)).

yes

The NFA in the example acceptsa*.
Another possibility is to give the regular expression and let the program construct the NFA thus:

sim \\- input("abaa") \- simula(thompson(("a":"b"):"a"*)).

yes

For regular expressions that are not on the formr1:r2, wherer1 is possibly infinite, it is also possible to gen
erate strings generated by the regular expression:

sim \\- input(A) \- simula(thompson(("a":"b":"c")*)).

A = [];

A = [[]|_A];

A = "abc";

A = ["abc",[]|_A];

A = "abcabc";

...

Yet another way to use the knowledge of the system is to assume a number of regular expressions an
if any of these generates an NFA which accepts the input string.

sim \\- reg(_,"b"),reg(_,"a"),reg(_,("a")*),input("a") \-
simula(thompson(reg(_,Exp))).

Exp = "a";

Exp = ("a"*);

no

One way to read this is: “Assuming thatreg(_,"a"), reg(_,"b") andreg(_,("a")*) are regular
expressions and that we have the input string"a" , is it then possible to construct an NFA that accepts t
input string?”

We could also pose the query: “Is the input string"a" accepted by both the NFA’s generated by the re
ular expressionsaa* anda”.

sim \\- input("a") \- simula(thompson("a")),simula(thompson("a":("a")*)).

yes

Instead of giving the regular expressions as assumptions in the query we can put them in a definition
will give us a lexical analyser for the defined regular expressions. With the definition

reg(digits,(reg(digit,_))*).
reg(digit,("0"|"1"|"2"|"3"|"4"|"5"|"6"|"7"|"8"|"9")).
reg(relop,("="|"=":"<"|">")).

some possible queries are:

sim \\- input("23") \- simula(thompson(reg(N,_))).

N = digits;

no

sim \\- input("=")   \- simula(thompson(reg(N,_))).

N = relop;

no

4.4 About the rule definition

As mentioned before a GCLA II program consists of two parts thedefinitionand therule definition.To get
the answers we want from the definitions shown above we have to write some new rules and strateg
example we do not want to try thed_right/2 rule on terms that we know have no definition such as nu
bers or lists. We therefore introduce a special class of terms, which we calldata, and restrict the
d_right/2  rule so that it’s not applicable to these.

data(X):- var(X).
data(X):- number(X).
data(X):- X == [].
data(X):- nonvar(X), functor(X,'.',2).
data(X):- functor(X,nfa,3).

d_right(C,PT) <=
not(data(C)),
atom(C),
clause(C,B),
(PT -> (A \- B)) -> (A \- C).

Another special construct isnextchar/2 , which lacks definition since it’s a so calledconstructorthat has
its own inference rulenext_right/1

next_right(PT) <=
hdt(String,H,T),
(PT -> (I@[input(T)|R] \- C))
-> (I@[input(String)|R] \- next_char(H,C)).

5 Using the duality of the comma constructor

In this section we will present another approach how to implement Thompson’s construction in GC
The idea behind this implementation is to use the duality of the comma-constructor in GCA II; to the l
the object level sequent the comma is read disjunctively and to the right of the object level seque
comma is read conjunctively.

Suppose we want to pose queries like: “Find all regular expressions that contain the expression th
responds to a given NFA as a subexpression” and “Find all subexpressions to the regular expressio
corresponds to a given NFA”.

This can be accomplished by using just a small definition of Thompson’s construction and the dua
the comma constructor. We will see that using the definition on the left side of the object level seque
responds to the first query above and using the definition on the right side of the object level sequen
sponds to the second query.

5.1 The definition

If we want to use the clauses of the definition on both sides, then we cannot use the definition of Thom
construction given in the last section. The reason for this is that in the definition given in the last s
many of the clauses were evaluated functionally, that is, they were using the arrow-construct. But the
construct is not symmetric; the interpretation of the arrow-construct on the left side of the sequent
from the interpretation of the arrow-construct on the right side of the sequent.

Instead we now let the different cases of Thompson’s construction be defined by simply listing the
sitions that the resulting NFA consists of. This means that we represent an NFA by just a comma-sep
list of transitions. We do not bother listing the states, symbols, start state and accepting state of the
since an NFA constructed by Thompson’s construction is uniquely defined by its transition function
states are denoted by successor-arithmetic and the numbering of the states always gives the start 
smallest number and the accepting state the largest number. For example, the NFA corresponding
expression a* is represented by

(m(0,eps,s(0)),m(s(s(0)),eps,s(s(s(0)))),m(s(0),a,s(s(0))),
m(s(s(0)),eps,s(0)),m(0,eps,s(s(s(0)))))

Each clause in the definition corresponds directly to the step in Thompson’s construction (see Sectio
The complete definition is given by:

thompson(eps,S,s(S)) <=
m(S,eps,s(S)).

thompson(A,S,s(S))
#{A\=(_|_),A\=(_:_),A\=’*’(_),A\=’+’(_),A\=’?’(_),A\=eps,A\=p(_)} <=
m(S,A,s(S)).

thompson((A|B),S,s(T)) <=
m(S,eps,s(S)),
m(S,eps,s(T2)),
m(T2,eps,s(T)),
m(T,eps,s(T)),
less_than(T2,s(T)),
less_than(s(S),T2),
thompson(A,s(S),T2),
thompson(B,s(T2),T).

thompson((A:B),S,T) <=
less_than(T2,T),
less_than(S,T2),
thompson(A,S,T2),
thompson(B,T2,T).

thompson((A)*,S,s(T)) <=
m(S,eps,s(S)),
m(S,eps,s(T)),
m(T,eps,s(T)),
m(T,eps,s(S)),
thompson(A,s(S),T).

thompson((A)+,S,T) <=
thompson((A: A*),S,T).

thompson((A)?,S,T) <=
thompson((A|eps),S,T).

thompson(p(E),S,T) <=
thompson(E,S,T).

less_than(0,s(_)).
less_than(s(A),s(B)) <=

less_than(A,B).

For example, the case when the regular expression is in the form ofa|b, i.e.,a or b is defined by

thompson((A|B),S,s(T)) <=
m(S,eps,s(S)),
m(S,eps,s(T2)),
m(T2,eps,s(T)),
m(T,eps,s(T)),
less_than(T2,s(T)),
less_than(s(S),T2),
thompson(A,s(S),T2),
thompson(B,s(T2),T).

If m(S1, Sym, S2) is read: “There is a transition from stateS1 to stateS2 on symbolSym”, then this defini-
tion can be read: “The NFA resulting when applying Thompson’s construction to a regular expression
form of a|b when starting in stateS and ending in states(T) , consists of an empty transition from stateS
to states(S) , an empty transition from stateS to states(T2) , an empty transition from stateT2 to state
s(T) , an empty transition from stateT to states(T) , the transitions of the NFA that results when applyin
Thompson’s construction toa, starting in states(S) and ending in stateT2 and the transitions of the NFA
that results when applying Thompson’s construction tob, starting in states(T2) and ending in stateT,
whereT2 is a new state that lies betweenS ands(T ).

5.2 Some rules

We also need some rules and strategies to be able to use this definition the way we want. For exam
only want thed_right/2 andd_left/3 rules to be applicable to thethompson/3 andless_than/2

clauses. We also need a newaxiom/3 rule, since we only want theaxiom to be applicable to variables an
on m/3 clauses. Since we do not want theaxiom/3 rule to be applicable when we are following a path
the search-tree that we know will not lead to a solution, we have to restrict theaxiom/3 rule to be applicable
only to sequents where there are no non-evaluatedthompson/3  or less_than/2  clauses and no
false/0  clauses. The newaxiom/3  rule becomes

axiom(A,C,I) <=
\+functor(A,thompson,3), % Not applicable to thompson/3.
\+functor(C,thompson,3),
\+functor(A,less_than,2), % Not applicable to less_than/2.
\+functor(C,less_than,2),
\+functor(A,true,0), % Not applicable to true/0.
\+functor(C,true,0),
\+member(thompson(_,_,_),I), % No non-evaluated thompson/3
\+member(thompson(_,_,_),R), % in the sequent.
\+member(less_than(_,_),I), % No non-evaluated less_than/2
\+member(less_than(_,_),R), % in the sequent.
\+member(false,I), % No false/0 in the sequent.
\+member(false,R),
term(C),
term(A),
unify(C,A) -> (I@[A|R] \- C).

5.3 Queries

There are two forms of queries that we can pose:

1. thompson( E, S, T) \- m1, m2,..., mn.
This query should be read: “Find all regular expressionsE whose corresponding NFA (as given b
Thompson’s construction with start stateS and accepting stateT) containsat leastthe transitions
m1, m2,..., mn.” If m1, m2,..., mn is the definition of an NFA and if this NFA corresponds to th
regular expressionr, this is equivalent to: “Find all regular expressionsE which containr as a sub-
expression.”

2. m1, m2,..., mn \- thompson( E, S, T) .
This query should be read: “Find all regular expressionsE whose corresponding NFA (as given b
Thompson’s construction with start stateS and accepting stateT) containsat mostthe transitions
m1, m2,..., mn.” If m1, m2,..., mn is the definition of an NFA and if this NFA corresponds to th
regular expressionr, this is equivalent to: “Find all regular expressionsE which are a subexpression
of r.”

5.3.1 Using the definition to the left

If we want to know which NFA that results when applying Thompson’s construction to the regular ex
siona|b we can pose the query

gclar \\- thompson((a|b),0,_) \- X.

and the system will respond with

X = m(0,eps,s(0));

X = m(0,eps,s(s(s(0))));

X = m(s(s(0)),eps,s(s(s(s(s(0))))));

X = m(s(s(s(s(0)))),eps,s(s(s(s(s(0))))));

X = m(s(0),a,s(s(0));

X = m(s(s(s(0))),b,s(s(s(s(s(0))))));

no

that is, all transitions in the NFA corresponding to the expressiona|b.
Another query is

gclar \\- thompson((a*|b:c),0,_) \- m(_,eps,T),m(T,a,_).

It should be read: “Is there an empty transition to stateT from any state and a transition from stateT to any
state on symbola, in the NFA corresponding toa*|bc? The system will answer twice withT = s(s(0))
since there are two such empty transitions, one from state 1 and one from state 3.

The NFA that corresponds to the expressionab is given by the transitions:m(0,a,s(0))  and
m(s(0),b,s(s(0))) . If we pose the query

gclar \\- thompson(E,0,s(s(0))) \- m(0,a,s(0)),m(s(0),b,s(s(0))).

the system will respond with the single answerE = (a:b) since that is the only expression whose corr
sponding NFA contains exactly those transitions. If we instead pose the query

gclar \\- thompson(E,0,s(s(0))) \- m(_,a,_),m(_,b,_).

we will get the answersE = (a:b) andE = (b:a) since these are the only expressions whose co
sponding NFA’s start in state0, end in states(s(0)) and have transitions onat leastsymbola and symbol
b.

If we do not want to restrict ourselves to finding expressions whose corresponding NFA’s are of a c
length, we can pose the following query

less_than(T,s(s(s(s(0)))),thompson(E,0,T) \- m(_,a,_),m(_,b,_).

which should be read: “Find all expressionsE whose corresponding NFA’s are of length less than 4 and c
tains transitions onat least symbola and symbol b. The system will respond with

E = (a:b)
T = s(s(0));

E = (b:a)
T = s(s(0));

E = (eps:(a:b))
T = s(s(s(0)));

E = (eps:(b:a))
T = s(s(s(0)));

...

If we instead we instead pose the query

less_than(T,s(s(s(s(0)))),thompson(E,0,T) \- m(_,a,N),m(N,b,_).

which should be read: “Find all expressionsE whose corresponding NFA’s are of length less than 4 and
whichab is a subexpression.” We will then get the answers

E = (a:b)
N = s(0)
T = s(s(0));

E = (eps:(a:b))
N = s(s(0))
T = s(s(s(0)));

...

5.3.2 Using the definition to the right

If we for some reason want to find all expressions whose corresponding NFA’s consist only of one tran
we can pose the query

m(_,X,_) \- thompson(E,0,_).

and the system will respond with

X = eps
E = eps;

X = X
E = _X;

no

that is,E is eithereps  or a variable.



ains
The query

m(_,a,_),m(_,b,_) \- less_than(T,s(s(s(0)))),thompson(E,0,T).

should be read: “Find all expressionsE whose corresponding NFA’s are of length less than 3 and cont
transitions onat mostsymbola and symbolb. We will get the following answers

E = a
T = s(0);

E = b
T = s(0);

E = (a:b)
T = s(s(0));

E = (b:a)
T = s(s(0));

no

If we instead pose the query

m(_,a,N),m(N,b,_) \- less_than(T,s(s(s(0)))),thompson(E,0,T).

we will get the answers

E = a
N = s(0)
T = s(0);

E = b
N = 0
T = s(0);

E = (a:b)
N = s(0)
T = s(s(0));

no

that is, we get all subexpressions ofab.

